[bookmark: _Hlk445622]/*The SAS program (HEI-2010 Individual Scores per day using ASA24-2011 to ASA24-2014 data)
                                                     ByPerson.SAS*/	
		     
/*This SAS program, along with the macros, can be used to calculate Healthy Eating Index (HEI)-2010 scores from 24-hour recall data collected using ASA24-2011, ASA24-Kids-2013, ASA24-2014 and ASA24-Kids-2014. This program calculates HEI-2010 component and total scores for each individual (using multiple ASA24 recalls, if available, for a single respondent). */
							      
/*This program has been tested using SAS, version 9.2 and uses the INFMYPHEI and TNMYPHEI analysis files from ASA24.  These files can be downloaded from the ASA24 Researcher website page. If any changes are made to the INFMYPHEI analysis file during data cleaning, please recreate an updated TNMYPHEI file prior to running this code. The files should be in CSV format. */

/*Please see accompanying readme file. */
**********************************************************************

[bookmark: _Hlk644889]/*INSTRUCTIONS – complete tasks 1-4 in this section, and run these SAS codes before proceeding to the HEI-2010 scoring program that follows*/

[bookmark: _Hlk349486][bookmark: _Hlk274099]/*1. Create a folder on your computer “home folder”, and save the ASA24 data, and the required HEI-2010 macros in it. Specify the path to the folder. */


%let home = C:\Users\Documents\ASA24; /*In this Example, the “home” folder is in C Drive, within Documents, and is called ASA24*/


[bookmark: _Hlk274158]/*2. Filenames here specifies the input files. */

[bookmark: _Hlk274320]filename inf4 “&home\inf4\inf4.xls”; 
[bookmark: _Hlk785054][bookmark: _GoBack]filename tn4 “&home\tn4\tn4.xls”; /*In this Example, the ASA24 Individual Foods data “inf4”, and the Total Foods data “tn4”, are in folders called “inf4” and “tn4”, all saved within the “home” folder. The data are in Excel format. */


[bookmark: _Hlk445500]/*3. Create a folder in the "home" folder, where the output file, containing HEI-2010 component and total scores for each respondent are to be exported. Specify the name of the folder. */

filename res “&home\res”; /*In this Example, the folder is called “RES”, within the “home” folder, and the exported results will be a csv file called “hei2010r”. */


[bookmark: _Hlk274641]/*4. Read in required HEI-2010 scoring macros. These macros must be saved within the home folder. */

%include “&home\hei2010.beanspeas.allocation.macro.sas”;
%include “&home\hei2010.score.macro.sas”;

[bookmark: _Hlk274947]/*NOTE: Once you have completed all the steps above, all you need to do is run the SAS program below. Unless you used different names for your dataset and folders, no other action is required from you.*/

TITLE 'ASA24 HEI-2010 scores - by person using all days';

/*Step 1. 
 Moves soy beverages out of Soybean Products (M_SOY) and into Dairy (D_TOTAL).
  This uses the ASA24 INFMYPHEI output file.  The adjusted totals are generated for
  each day of dietary intake and saved in a temporary SAS dataset.
 */

*Input the food data;
proc import datafile=inf4
  out=inf4
  dbms=xls
  replace;
  getnames=yes;
run;

data food;
  set inf4;

  *move soy milk from soy to dairy;
  /*FOODCODE=11310000, MILK, IMITATION, FLUID, SOY BASED (1 cup=244 grams)
    FOODCODE=11320000, MILK, SOY, READY-TO-DRINK, NOT BABY (1 cup=245 grams)
    FOODCODE=11321000, MILK, SOY, READY-TO-DRINK, NOT BABY'S, CHOCOLATE (1 cup=240 grams)
    FOODCODE=11330000, MILK, SOY, DRY, RECONSTITUTED, NOT BABY (1 cup=245 grams) */
  IF FOODCODE=11310000 THEN DO;
     M_SOY=0;
	 D_TOTAL=ROUND(FoodAmt/244,.001);
	 END;
  ELSE IF FOODCODE=11320000 THEN DO;
     M_SOY=0;
	 D_TOTAL=ROUND(FoodAmt/245,.001);
	 END;
  ELSE IF FOODCODE=11321000 THEN DO;
     M_SOY=0;
	 D_TOTAL=ROUND(FoodAmt/240,.001);
	 END;
  ELSE IF FOODCODE=11330000 THEN DO;
     M_SOY=0;
	 D_TOTAL=ROUND(FoodAmt/245,.001);
	 END;


run;

proc sort data=food;
  by username recallno;
run;

proc means data=food noprint;
  by username recallno;
  var d_total m_soy;
  output out=adjsum sum=d_total_adj m_soy_adj;
run;

/*Step 2. 
 Merges the TNMYPHEI file with the SAS dataset created in Step 1, creating a file with
  adjusted variables for the MPED variables of Total Milk (D_TOTAL_ADJ) and Soybean Products (M_SOY_ADJ).
*/ 

*Input the day total data;
proc import datafile=tn4
  out=tn4
  dbms=xls
  replace;
  getnames=yes;
run;

proc sort data=tn4;
  by username recallno;
  
data tn4a;
  merge tn4 adjsum;
  by username recallno;
run;

/*Step 3.
 Creates four additional required variables.  These variables are:  MONOPOLY, ALLMEAT, SEAPLANT, and EMPTYCAL10. 
*/


DATA tn4a;
  SET tn4a;
  by username recallno;

  MONOPOLY=mfat+pfat;

  ALLMEAT=M_MPF+M_EGG+M_NUTSD+m_soy_adj;
  SEAPLANT=M_FISH_HI+M_FISH_LO+m_soy_adj+M_NUTSD;

 /**Calculate intake of Calories from SoFAAS**/

      ADDSUGC=16*ADD_SUG;	/*calories from added sugars*/
      SOLFATC=DISCFAT_SOL*9;	/*calories from solid fat*/

      maxalcgr=13*(kcal/1000);  /*max grams of alcohol based on kcal intake*/

      if ALC <= maxalcgr then EXALCCAL=0;  /*consumed less than max*/
        else if ALC > maxalcgr then EXALCCAL=7*(ALC-maxalcgr);  /*get cal from extra alc grams*/

      EMPTYCAL10=ADDSUGC+SOLFATC+EXALCCAL; /*total empty calories in hei2010 definition*/


                           
run; 

/*Step 4.
 Calculates total food group and nutrient intake over all possible days reported per individual.
*/

*get sum per person of variables of interest;
proc means data=tn4a noprint;
  by username;
  var KCAL allmeat seaplant v_total v_drkgr legumes f_total wholefrt g_whl d_total_adj
     monopoly sfat SODI G_NWHL EMPTYCAL10;
  output out=persumtn4 sum=;
run;


/*Step 5.
 Runs the macro to properly allocate the intakes of the MPED variable Legumes (LEGUMES)
  to either Total Protein Foods and Seafood and Plant Proteins (ALLMEAT and SEAPLANT)
  or Total Vegetables and Greens and Beans (V_TOTAL and V_DRKGR).  The four resulting
  variables from this step, named LEGUME_ADDED_V_TOTAL, LEGUME_ADDED_BEANGRN,
  LEGUME_ADDED_ALLMEAT, and LEGUME_ADDED_SEAPLANT are all used in the next step. 
*/

%LEG2010A (indat=persumtn4,
           kcal=KCAL,
           allmeat=allmeat,
	   seaplant=seaplant,
	   v_total=v_total,
	   v_drkgr=v_drkgr,
	   legumes=legumes,
	   outdat=afterleg);

/*Step 6. 
 Runs the HEI2010 scoring macro which calculates intake density amounts and HEI scores.
*/

%HEI2010 (indat=afterleg,
          kcal=KCAL,
	  lv_total=legume_added_V_TOTAL,
	  lbeangrn=legume_added_BEANGRN,
	  f_total=f_total,
	  wholefrt=wholefrt
	  g_whl=g_whl,
	  d_total=d_total_adj,
          lallmeat=legume_added_ALLMEAT,
	  lseaplant=legume_added_SEAPLANT,
	  monopoly=monopoly,
	  sfat=sfat,
	  sodi=SODI,
	  G_NWHL=G_NWHL,
	  EMPTYCAL10=EMPTYCAL10,
	  outdat=hei2010);
 
run;

/*Step 7. 
 Displays and saves the results.
*/ 


Data hei2010r (keep=username HEIX1_TOTALVEG HEIX2_GREEN_AND_BEAN HEIX3_TOTALFRUIT
     HEIX4_WHOLEFRUIT HEIX5_WHOLEGRAIN HEIX6_TOTALDAIRY HEIX7_TOTPROT HEIX8_SEAPLANT_PROT
     HEIX9_FATTYACID HEIX10_SODIUM HEIX11_REFINEDGRAIN HEIX12_SOFAAS
     HEI2010_TOTAL_SCORE);
  Set hei2010;
  Run;

proc means n nmiss min max mean data=hei2010r;
run;


proc export data= hei2010r
  file=res
  dbms=csv
  replace;
run;

